First-Order Linear Equations and Sensitive Initial Conditions

Example 1. We look at the first order linear differential equation

$$\frac{dx}{dt} = 10\,t - 2\,tx.$$

```
> restart:with(plots):with(DEtools):
```

We enter the differential equation.

```
> deq:= diff(y(x),x)=10*x-2*x*y(x);
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}x}\,y(x) = 10\,x - 2\,x\,y(x)$$

We find the general solution.

```
> soln:= dsolve(deq,y(x));
```

$$soln := y(x) = 5 + \mathrm{e}^{-x^2}\,\_C1$$

We now find the solutions for the initial value problems corresponding to the initial conditions $y(0) = -20$, $y(0) = -15$, $y(0) = -10$, $y(0) = -5$, $y(0) = 0$, $y(0) = 5$, $y(0) = 10$, $y(0) = 15$, $y(0) = 20$, $y(0) = 25$, and $y(0) = 30$, respectively.

```
> for i from -4 to 6 do
  IC[i]:=y(0)=5*i;
  solution[i]:=dsolve({deq,IC[i]},y(x));
  printf("\n"):
  od;
```

$$IC_{-4} := y(0) = -20$$

$$solution_{-4} := y(x) = 5 - 25\,\mathrm{e}^{-x^2}$$

$$IC_{-3} := y(0) = -15$$

$$solution_{-3} := y(x) = 5 - 20\,\mathrm{e}^{-x^2}$$

$$IC_{-2} := y(0) = -10$$

$$solution_{-2} := y(x) = 5 - 15\,\mathrm{e}^{-x^2}$$

$$IC_{-1} := y(0) = -5$$

$$solution_{-1} := y(x) = 5 - 10\,\mathrm{e}^{-x^2}$$

$$IC_0 := y(0) = 0$$

$$solution_0 := y(x) = 5 - 5\,e^{-x^2}$$

$$IC_1 := y(0) = 5$$

$$solution_1 := y(x) = 5$$

$$IC_2 := y(0) = 10$$

$$solution_2 := y(x) = 5 + 5\,e^{-x^2}$$

$$IC_3 := y(0) = 15$$

$$solution_3 := y(x) = 5 + 10\,e^{-x^2}$$

$$IC_4 := y(0) = 20$$

$$solution_4 := y(x) = 5 + 15\,e^{-x^2}$$

$$IC_5 := y(0) = 25$$

$$solution_5 := y(x) = 5 + 20\,e^{-x^2}$$

$$IC_6 := y(0) = 30$$

$$solution_6 := y(x) = 5 + 25\,e^{-x^2}$$

Now we plot the direction field for the differential equation along with the 11 solutions for different values of the initial conditions.

```
> DEplot (deq, y(x), x = 0..3, y=-20..30,{seq([0,5*i],i=-4..6)},
  arrows = medium, color=magenta, linecolor=green);
```



Notice how all the solutions approach the constant solution $y = 5$ no matter what the initial condition. We refer to the terms containing e^{-t^2} in the solution as the **transient** part of the solution since it approaches 0 exponentially as t approaches *infinity*. The term 5 in each solution is referred to as the **steady state** part of the solution. The ultimate process is not sensitive to our choice of initial conditions (or constants).

Example 2. We now look at another differential equation, although not linear.

```
> deq2:= diff(y(x),x)=y(x)^2-4;
```

$$deq2 := \frac{d}{dx}\, y(x) = y(x)^2 - 4$$

We see this differential equation has $x = 2$ and $x = -2$ as constant solutions. We find the general solution.

```
> soln2:= dsolve(deq2,y(x));
```

$$soln2 := y(x) = -\frac{2\left(e^{4x}\_C1 + 1\right)}{-1 + e^{4x}\_C1}$$

We look at the direction field for this differential equation.

```
> DEplot (deq2, y(x), x = 0..3, y=-6..6, arrows = medium, color=
  magenta);
```



We now plot the direction field again along with the solutions corresponding to the initial conditions $y(0) = -6$, $y(0) = -5$, $y(0) = -4$, $y(0) = -3$, $y(0) = -2.01$, $y(0) = -2$, $y(0) = -1.99$, $y(0) = -1$, $y(0) = 0$, $y(0) = 1$, $y(0) = 1.99$, $y(0) = 2$, $y(0) = 2.01$, $y(0) = 3$, $y(0) = 4$, $y(0) = 5$, and $y(0) = 6$.

```
> DEplot (deq2, y(x), x = 0..3, y=-6..6, {[0,-6],[0,-5],[0,-4],[0,
  -3],[0,-2.01],[0,-2],[0,-1.99],[0,-1],[0,0],[0,1],[0,1.99],[0,2],
  [0,2.01],[0,3],[0,4],[0,5],[0,6]},arrows = medium, color=magenta,
  linecolor=green);
```



Here we see that all solutions for initial conditions y(0) < 2 converge to the constant solution $y=-2$. On the other hand, all solutions for initial conditions y(0) > 2 go off to +*infinity*. **We have a situation where the ultimate behavior of a differential equation is very sensitive to some very small changes in initial condition.** Let's look at the plot again for the three very close initial conditions $y(0) = 1.99$ (the solution ultimately converges to the constant solution $y(t) = -2$), for $y(0) = 2$ (we have the constant solution $y = 2$), and $y(0) = 2.01$ (the solution quickly moves off toward infinity).

```
> DEplot (deq2, y(x), x = 0..3, y=-6..6, {[0,-2],[0,1.99],[0,2],[0,
  2.01]},arrows = medium, color=magenta, linecolor=[green,red,yellow,
  blue]);
```



6
4
2
0
−2
−4
−6
y(x)
1
2
3
x
Curve 1
Curve 2
Curve 3
Curve 4
Curve 5